

sdtest — Variance-comparison tests

| | | | |
|-----------------------------|--------------------------------------|--------------------------------|--------------------------------------|
| Description | Quick start | Menu | Syntax |
| Options | Remarks and examples | Stored results | Methods and formulas |
| References | Also see | | |

Description

`sdtest` performs tests on the equality of standard deviations (variances). In the first form, `sdtest` tests that the standard deviation of *varname* is *#*. In the second form, `sdtest` performs the same test, using the standard deviations of the two groups defined by *groupvar*. In the third form, `sdtest` tests that *varname*₁ and *varname*₂ have the same standard deviation.

`sdtesti` is the immediate form of `sdtest`; see [\[U\] 19 Immediate commands](#).

Both the traditional F test for the homogeneity of variances and Bartlett's generalization of this test to K samples are sensitive to the assumption that the data are drawn from an underlying Gaussian distribution. See, for example, the cautionary results discussed by [Markowski and Markowski \(1990\)](#). [Levene \(1960\)](#) proposed a test statistic for equality of variance that was found to be robust under nonnormality. Then [Brown and Forsythe \(1974\)](#) proposed alternative formulations of Levene's test statistic that use more robust estimators of central tendency in place of the mean. These reformulations were demonstrated to be more robust than Levene's test when dealing with skewed populations.

`robvar` reports Levene's robust test statistic (W_0) for the equality of variances between the groups defined by *groupvar* and the two statistics proposed by Brown and Forsythe that replace the mean in Levene's formula with alternative location estimators. The first alternative (W_{50}) replaces the mean with the median. The second alternative replaces the mean with the 10% trimmed mean (W_{10}).

Quick start

Test that the standard deviation of *v1* is equal to 2

```
sdtest v1=2
```

Equality of standard deviations (variances) test for *v1* comparing the two groups defined by *catvar1*

```
sdtest v1, by(catvar1)
```

Robust equality of variances test for *v1* comparing the groups defined by *catvar1*

```
robvar v1, by(catvar1)
```

Compare the variances of *v2* and *v3*

```
sdtest v2 == v3
```

Same as above, but with separate tests for each group defined by *catvar2*

```
by catvar2, sort: sdtest v2 == v3
```

Test $sd_1 = sd_2$ for $sd_1 = 34$, $sd_2 = 45$, $N_1 = 143$, and $N_2 = 184$

```
sdtesti 143 . 34 184 . 45
```

Menu

sdtest

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > Variance-comparison test

sdtesti

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > Variance-comparison test calculator

robvar

Statistics > Summaries, tables, and tests > Classical tests of hypotheses > Robust equal-variance test

Syntax

One-sample variance-comparison test

```
sdtest varname == # [if] [in] [, level(#)]
```

Two-sample variance-comparison test using groups

```
sdtest varname [if] [in] , by(groupvar) [level(#)]
```

Two-sample variance-comparison test using variables

```
sdtest varname1 == varname2 [if] [in] [, level(#)]
```

Immediate form of one-sample variance-comparison test

```
sdtesti #obs {#mean | .} #sd #val [, level(#)]
```

Immediate form of two-sample variance-comparison test

```
sdtesti #obs,1 {#mean,1 | .} #sd,1 #obs,2 {#mean,2 | .} #sd,2 [, level(#)]
```

Robust tests for equality of variances

```
robvar varname [if] [in] , by(groupvar)
```

by and collect are allowed with `sdtest` and `robvar`, and collect is allowed with `sdtesti`; see [\[U\] 11.1.10 Prefix commands](#).

Options

`level(#)` specifies the confidence level, as a percentage, for confidence intervals of the means. The default is `level(95)` or as set by `set level`; see [\[U\] 20.8 Specifying the width of confidence intervals](#).

`by(groupvar)` specifies the *groupvar* that defines the groups to be compared. For `sdtest`, there should be two groups, but for `robvar` there may be more than two groups. Do not confuse the `by()` option with the `by` prefix; both may be specified.

Remarks and examples

Remarks are presented under the following headings:

Basic form
Immediate form
Robust test

Basic form

`sdtest` performs two different statistical tests: one testing equality of variances and the other testing that the standard deviation is equal to a known constant. Which test it performs is determined by whether you type a variable name or a number to the right of the equal sign.

► Example 1: One-sample test of variance

We have a sample of 74 automobiles. For each automobile, we know the mileage rating. We wish to test whether the overall standard deviation is 5 mpg:

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
```

```
. sdtest mpg == 5
```

One-sample test of variance

| Variable | Obs | Mean | Std. err. | Std. dev. | [95% conf. interval] | |
|----------|-----|---------|-----------|-----------|----------------------|----------|
| mpg | 74 | 21.2973 | .6725511 | 5.785503 | 19.9569 | 22.63769 |

```
sd = sd(mpg)                                c = chi2 = 97.7384
H0: sd = 5                                  Degrees of freedom = 73
Ha: sd < 5                                Ha: sd != 5                Ha: sd > 5
Pr(C < c) = 0.9717                          2*Pr(C > c) = 0.0565      Pr(C > c) = 0.0283
```

◀

► Example 2: Variance ratio test

We are testing the effectiveness of a new fuel additive. We run an experiment on 12 cars, running each without and with the additive. The data can be found in [R] `tttest`. The results for each car are stored in the variables `mpg1` and `mpg2`:

```
. use https://www.stata-press.com/data/r18/fuel
```

```
. sdtest mpg1==mpg2
```

Variance ratio test

| Variable | Obs | Mean | Std. err. | Std. dev. | [95% conf. interval] | |
|----------|-----|--------|-----------|-----------|----------------------|----------|
| mpg1 | 12 | 21 | .7881701 | 2.730301 | 19.26525 | 22.73475 |
| mpg2 | 12 | 22.75 | .9384465 | 3.250874 | 20.68449 | 24.81551 |
| Combined | 24 | 21.875 | .6264476 | 3.068954 | 20.57909 | 23.17091 |

```
ratio = sd(mpg1) / sd(mpg2)                                f = 0.7054
H0: ratio = 1                                              Degrees of freedom = 11, 11
Ha: ratio < 1                                Ha: ratio != 1                Ha: ratio > 1
Pr(F < f) = 0.2862                          2*Pr(F < f) = 0.5725      Pr(F > f) = 0.7138
```

We cannot reject the hypothesis that the standard deviations are the same.

In [R] `ttest`, we draw an important distinction between paired and unpaired data, which, in this example, means whether there are 12 cars in a before-and-after experiment or 24 different cars. For `sdtest`, on the other hand, there is no distinction. If the data had been unpaired and stored as described in [R] `ttest`, we could have typed `sdtest mpg, by(treated)`, and the results would have been the same.



Immediate form

Example 3: `sdtesti`

Immediate commands are used not with data, but with reported summary statistics. For instance, to test whether a variable on which we have 75 observations and a reported standard deviation of 6.5 comes from a population with underlying standard deviation 6, we would type

```
. sdtesti 75 . 6.5 6
One-sample test of variance
```

| | Obs | Mean | Std. err. | Std. dev. | [95% conf. interval] |
|---|-----|------|-----------|-----------|----------------------|
| x | 75 | . | .7505553 | 6.5 | . . |

```

sd = sd(x)
HO: sd = 6
Ha: sd < 6
Pr(C < c) = 0.8542

c = chi2 = 86.8472
Degrees of freedom = 74
Ha: sd != 6
2*Pr(C > c) = 0.2916
Ha: sd > 6
Pr(C > c) = 0.1458
```

The mean plays no role in the calculation, so it may be omitted.

To test whether the variable comes from a population with the same standard deviation as another for which we have a calculated standard deviation of 7.5 over 65 observations, we would type

```
. sdtesti 75 . 6.5 65 . 7.5
Variance ratio test
```

| | Obs | Mean | Std. err. | Std. dev. | [95% conf. interval] |
|----------|-----|------|-----------|-----------|----------------------|
| x | 75 | . | .7505553 | 6.5 | . . |
| y | 65 | . | .9302605 | 7.5 | . . |
| Combined | 140 | . | . | . | . . |

```

ratio = sd(x) / sd(y)
HO: ratio = 1
Ha: ratio < 1
Pr(F < f) = 0.1172

f = 0.7511
Degrees of freedom = 74, 64
Ha: ratio != 1
2*Pr(F < f) = 0.2344
Ha: ratio > 1
Pr(F > f) = 0.8828
```



Robust test

► Example 4: robvar

We wish to test whether the standard deviation of the length of stay for patients hospitalized for a given medical procedure differs by gender. Our data consist of observations on the length of hospital stay for 1778 patients: 884 males and 894 females. Length of stay, `lengthstay`, is highly skewed (skewness coefficient = 4.912591) and thus violates Bartlett's normality assumption. Therefore, we use `robvar` to compare the variances.

```
. use https://www.stata-press.com/data/r18/stay
. robvar lengthstay, by(sex)
```

| Gender | Summary of Length of stay in days | | |
|--------|-----------------------------------|-----------|-------|
| | Mean | Std. dev. | Freq. |
| Male | 9.0874434 | 9.7884747 | 884 |
| Female | 8.800671 | 9.1081478 | 894 |
| Total | 8.9432508 | 9.4509466 | 1,778 |

W0 = 0.55505315 df(1, 1776) Pr > F = 0.45635888
W50 = 0.42714734 df(1, 1776) Pr > F = 0.51347664
W10 = 0.44577674 df(1, 1776) Pr > F = 0.50443411

For these data, we cannot reject the null hypothesis that the variances are equal. However, Bartlett's test yields a significance probability of 0.0319 because of the pronounced skewness of the data. ◀

□ Technical note

`robvar` implements both the conventional Levene's test centered at the mean and a median-centered test. In a simulation study, [Conover, Johnson, and Johnson \(1981\)](#) compare the properties of the two tests and recommend using the median test for asymmetric data, although for small sample sizes the test is somewhat conservative. See [Carroll and Schneider \(1985\)](#) for an explanation of why both mean- and median-centered tests have approximately the same level for symmetric distributions, but for asymmetric distributions the median test is closer to the correct level. □

Stored results

`sdtest` and `sdtesti` store the following in `r()`:

Scalars

| | |
|----------------------|--|
| <code>r(N)</code> | number of observations |
| <code>r(p_l)</code> | lower one-sided p -value |
| <code>r(p_u)</code> | upper one-sided p -value |
| <code>r(p)</code> | two-sided p -value |
| <code>r(F)</code> | F statistic |
| <code>r(sd)</code> | standard deviation |
| <code>r(sd_1)</code> | standard deviation for first variable |
| <code>r(sd_2)</code> | standard deviation for second variable |
| <code>r(df)</code> | degrees of freedom |
| <code>r(df_1)</code> | numerator degrees of freedom |
| <code>r(df_2)</code> | denominator degrees of freedom |
| <code>r(chi2)</code> | χ^2 |

`robvar` stores the following in `r()`:

Scalars

| | |
|-----------------------|---|
| <code>r(N)</code> | number of observations |
| <code>r(w50)</code> | Brown and Forsythe's F statistic (median) |
| <code>r(p_w50)</code> | Brown and Forsythe's p -value |
| <code>r(w0)</code> | Levene's F statistic |
| <code>r(p_w0)</code> | Levene's p -value |
| <code>r(p_w10)</code> | Brown and Forsythe's F statistic (trimmed mean) |
| <code>r(p_w10)</code> | Brown and Forsythe's p -value (trimmed mean) |
| <code>r(df_1)</code> | numerator degrees of freedom |
| <code>r(df_2)</code> | denominator degrees of freedom |

Methods and formulas

See Armitage et al. (2002, 149–153) or Bland (2015, 144–145) for an introduction and explanation of the calculation of these tests.

The test for $\sigma = \sigma_0$ is given by

$$\chi^2 = \frac{(n-1)s^2}{\sigma_0^2}$$

which is distributed as χ^2 with $n-1$ degrees of freedom.

The test for $\sigma_x^2 = \sigma_y^2$ is given by

$$F = \frac{s_x^2}{s_y^2}$$

which is distributed as F with $n_x - 1$ and $n_y - 1$ degrees of freedom.

Let X_{ij} be the j th observation of X for the i th group. Let $Z_{ij} = |X_{ij} - \bar{X}_i|$, where \bar{X}_i is the mean of X in the i th group. Levene's test statistic is

$$W_0 = \frac{\sum_i n_i (\bar{Z}_i - \bar{Z})^2 / (g-1)}{\sum_i \sum_j (Z_{ij} - \bar{Z}_i)^2 / \sum_i (n_i - 1)}$$

where n_i is the number of observations in group i and g is the number of groups. W_{50} is obtained by replacing \bar{X}_i with the i th group median of X_{ij} , whereas W_{10} is obtained by replacing \bar{X}_i with the 10% trimmed mean for group i .

References

- Armitage, P., G. Berry, and J. N. S. Matthews. 2002. *Statistical Methods in Medical Research*. 4th ed. Oxford: Blackwell.
- Bland, M. 2015. *An Introduction to Medical Statistics*. 4th ed. Oxford: Oxford University Press.
- Brown, M. B., and A. B. Forsythe. 1974. Robust tests for the equality of variances. *Journal of the American Statistical Association* 69: 364–367. <https://www.jstor.org/stable/2285659>.
- Carroll, R. J., and H. Schneider. 1985. A note on Levene's tests for equality of variances. *Statistics and Probability Letters* 3: 191–194. [https://doi.org/10.1016/0167-7152\(85\)90016-1](https://doi.org/10.1016/0167-7152(85)90016-1).
- Conover, W. J., M. E. Johnson, and M. M. Johnson. 1981. A comparative study of tests for homogeneity of variances, with applications to the outer continental shelf bidding data. *Technometrics* 23: 351–361. <https://doi.org/10.2307/1268225>.
- Gastwirth, J. L., Y. R. Gel, and W. Miao. 2009. The impact of Levene's test of equality of variances on statistical theory and practice. *Statistical Science* 24: 343–360. <https://doi.org/10.1214/09-STS301>.

- Levene, H. 1960. Robust tests for equality of variances. In *Contributions to Probability and Statistics: Essays in Honor of Harold Hotelling*, ed. I. Olkin, S. G. Ghurye, W. Hoeffding, W. G. Madow, and H. B. Mann, 278–292. Menlo Park, CA: Stanford University Press.
- Markowski, C. A., and E. P. Markowski. 1990. Conditions for the effectiveness of a preliminary test of variance. *American Statistician* 44: 322–326. <https://doi.org/10.2307/2684360>.

Also see

[R] `ttest` — t tests (mean-comparison tests)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

